
1. [bookmark: _Toc122434485][bookmark: _Toc107526170]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D12wk23 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc107526171]Corrections required
1.1 [bookmark: _Toc107494143][bookmark: _Toc107526172]Change 1
	Function name
	fl_TC_11_4_4_NR5GC_TestBody

	Reason for change
	1. Coordination triggers are missing after f_NR5GC_EmergencyCallPDUSession() to indicate to IMS PTC that PDU session modification to add voice has been done.
Note : IMS PTC is waiting for this trigger inside f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(). The purpose is similar to the change made by R5s220821 in f_NR5GC_EmergencyCallWithRegistration(), which is not used in this test case.

2. At step 12, there are no DRBs configured at the UE so the PDU session release procedure is done without any associated RRCReconfiguration. But, it is still required to delete the stored information about DRBs associated with this PDU session. Currently it is observed that if the UE re-uses the PDU session ID of a session released at step 12 for the next emergency PDU session, at the end of the emergency call the TTCN may find an additional associated DRB to release which is not configured at the SS, which can cause an error in SS operation.

	Summary of change
	1. Added a coordination messages at steps 8A-9D and 16G-16L, to be sent to the IMS PTC indicate call setup complete from NR5GC side

2. Following PDU session release at step 12, find and delete stored info for DRBs associated with that PDU session, and also delete stored info for the PDU session itself.

	TTCN module
	Emergency_NR5GC.ttcn

	MCC160 Comment
	1. Not needed – this trigger is added into f_NR5GC_EmergencyCallPDUSession instead.  See comments for R5s220821.
2. Accepted. Fixed implementation issue



Before Change:
	function fl_TC_11_4_4_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var O1_Type v_PDUSessionIdToRelease;
    var PDUSessionInfoList_Type v_PDUSessionInfo;
    var integer i;
    var integer v_NoOfSessions;
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var template (omit) PDN_Index_Type v_PDN_Index := omit; //@sic R5-220042 sic@

    // @siclog "Step 1" siclog@
    // The SS configures:
    //  - NR Cell 11 as "Serving cell"
    //  - NR Cell 1 as "Non-Suitable "Off" cell".

    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);
    f_NR_SetCellPower (nr_Cell11, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);
    
    // @siclog "Steps 2 - 4" siclog@
    // The UE transmits an RRCSetupRequest message.
    // SS transmit an RRCSetup message.
    // The UE transmits an RRCSetupComplete message to confirm the successful completion of the connection establishment and
    // a REGISTRATION REQUEST message indicating "mobility registration updating" is sent to update the registration of the actual tracking area.
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
    //@siclog "Step 5" siclog@
    // SS sends a REGISTRATION REJECT message containing 5GMM cause value = #22 (Congestion).
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_NoSecurityProtection, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_Congestion), cs_GPRS_Timer2_3_IEI('5F'O, '001'B, '00011'B)))));
    
    //@siclog "Step 6" siclog@
    // The SS transmits an RRCConnectionRelease message.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 7" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    f_UT_RequestIMSEmergencyCall(UT, px_EmergencyCallNumber);
    
    //@siclog "Step 8" siclog@
    //@sic R5-213508 sic@
    //Check: Does the UE perform steps 1-3 from the Registration procedure for mobility registration update as specified in TS 38.508-1 [4] subclause 4.9.5
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) { //@sic R5-213508 sic@
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
    //@sic R5-213508 sic@
    //@siclog "Step 8AA - 8AB" siclog@
    //The SS transmits a SecurityModeCommand message.
    //The UE transmits a SecurityModeComplete message.
    f_NR_RRC_ActivateSecurity (nr_Cell11, valueof(v_ReceivedMsg.SecurityProtection.NasCount));
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease);
    
    // @siclog "Step 8A - 9D" siclog@
    //Check: Does the UE transmit an UL NAS TRANSPORT message with 'Request type' set to 'initial emergency request', and, a PDU SESSION ESTABLISHMENT REQUEST for establishing an emergency PDU?
    // Steps 9-13 from Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11, including the parallel behaviour specified in table 4.9.11.2.2-2 are performed.
    //f_NR5GC_EmergencyCallWithRegistration(nr_Cell11);
    f_NR5GC_EmergencyCallPDUSession(nr_Cell11, tsc_NR_RbId_SRB1); //@sic R5s210515 sic@
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    // Emergency call now active
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8A");
    
    // @siclog "Step 10 - 11" siclog@
    // Make the UE release the emergency call. (NOTE 1)
    f_NR5GC_EmergencyCallRelease(nr_Cell11);
    
    // @sic R5s211524 sic@

    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_NoOfSessions := f_NR5GC_MobileInfo_GetNoOfSessions();
    // @siclog "Step 12" siclog@
    for (i := 0; i<v_NoOfSessions; i:=i+1)
    {
      // @sic R5s210515 R5-216200 sic@
      v_PDUSessionIdToRelease := v_PDUSessionInfo[0].SessionId; // @sic R5s220614 sic@
      v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease,
                                                       cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                                                       cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00101'B)); //T3396 = 5 minutes //                                                                p_NGSM_CongestionReattemptInd,
      SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB2, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));//@sic R5s210808sic@
      SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11,
                                        tsc_NR_RbId_SRB2, //@sic R5s210808sic@
                                        cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
      v_PDN_Index := f_Get_PDN_IndexForPDUSessionId (v_PDUSessionInfo, v_PDUSessionIdToRelease); //@sic R5-220042 sic@
      if (isvalue(v_PDN_Index)) {
       f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(valueof(v_PDN_Index)));
      } else {
       f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Cannot find PDN Index of PDU Session being released");
      }
      if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
      }
    }

    // @siclog "Step 13" siclog@
    // Void
    
    // @siclog "Step 14" siclog@
    // The SS releases the RRC connection.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 15" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    f_UT_RequestIMSEmergencyCall (UT, px_EmergencyCallNumber);
    
    // @sic R5-223350 sic@
    //@siclog "Steps 16A - 16E" siclog@
    //Check: Does the UE performs steps 1-5 of Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11?
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, emergency);
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell11)), '0011'B)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
   
    // RRC Security Mode pocedure
    f_NR_RRC_ActivateSecurity (nr_Cell11, v_ReceivedMsg.SecurityProtection.NasCount);
    
    //@siclog "Step 16F" siclog@
    //The SS transmits a SERVICE ACCEPT message.
    SRB.send (cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, f_Get_NG_ServiceAcceptMsg(v_ReceivedMsg.Pdu.Msg.service_Request.pduSessionStatus))));
   
    
    //@siclog "Steps 16G-16L" siclog@
    //Check: Does the UE perform steps 8-13 of Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11?
    // Steps 8-13
    f_NR5GC_EmergencyCallPDUSession(nr_Cell11, tsc_NR_RbId_SRB1);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "IMS Emergency Call Established");
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 16G-16L ");
 
    //@siclog "Step 17 - 18" siclog@
    //Make the UE release the emergency call. (NOTE 1)
    //The Generic test procedure for IMS MO Emergency call release as specified in TS 38.508-1 [4], subclause 4.9.12A takes place
    f_NR5GC_EmergencyCallRelease(nr_Cell11);

    f_NR_RRCRelease(nr_Cell11);
  } 




After Change:
	  function fl_TC_11_4_4_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var O1_Type v_PDUSessionIdToRelease;
    var PDUSessionInfoList_Type v_PDUSessionInfo;
    var integer i;
    var integer v_NoOfSessions;
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var template (omit) PDN_Index_Type v_PDN_Index := omit; //@sic R5-220042 sic@

    // @siclog "Step 1" siclog@
    // The SS configures:
    //  - NR Cell 11 as "Serving cell"
    //  - NR Cell 1 as "Non-Suitable "Off" cell".

    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);
    f_NR_SetCellPower (nr_Cell11, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);
    
    // @siclog "Steps 2 - 4" siclog@
    // The UE transmits an RRCSetupRequest message.
    // SS transmit an RRCSetup message.
    // The UE transmits an RRCSetupComplete message to confirm the successful completion of the connection establishment and
    // a REGISTRATION REQUEST message indicating "mobility registration updating" is sent to update the registration of the actual tracking area.
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
    //@siclog "Step 5" siclog@
    // SS sends a REGISTRATION REJECT message containing 5GMM cause value = #22 (Congestion).
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_NoSecurityProtection, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_Congestion), cs_GPRS_Timer2_3_IEI('5F'O, '001'B, '00011'B)))));
    
    //@siclog "Step 6" siclog@
    // The SS transmits an RRCConnectionRelease message.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 7" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    f_UT_RequestIMSEmergencyCall(UT, px_EmergencyCallNumber);
    
    //@siclog "Step 8" siclog@
    //@sic R5-213508 sic@
    //Check: Does the UE perform steps 1-3 from the Registration procedure for mobility registration update as specified in TS 38.508-1 [4] subclause 4.9.5
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) { //@sic R5-213508 sic@
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
    //@sic R5-213508 sic@
    //@siclog "Step 8AA - 8AB" siclog@
    //The SS transmits a SecurityModeCommand message.
    //The UE transmits a SecurityModeComplete message.
    f_NR_RRC_ActivateSecurity (nr_Cell11, valueof(v_ReceivedMsg.SecurityProtection.NasCount));
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease);
    
    // @siclog "Step 8A - 9D" siclog@
    //Check: Does the UE transmit an UL NAS TRANSPORT message with 'Request type' set to 'initial emergency request', and, a PDU SESSION ESTABLISHMENT REQUEST for establishing an emergency PDU?
    // Steps 9-13 from Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11, including the parallel behaviour specified in table 4.9.11.2.2-2 are performed.
    //f_NR5GC_EmergencyCallWithRegistration(nr_Cell11);
    f_NR5GC_EmergencyCallPDUSession(nr_Cell11, tsc_NR_RbId_SRB1); //@sic R5s210515 sic@
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    // Emergency call now active
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8A");
    
    // @siclog "Step 10 - 11" siclog@
    // Make the UE release the emergency call. (NOTE 1)
    f_NR5GC_EmergencyCallRelease(nr_Cell11);
    
    // @sic R5s211524 sic@

    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_NoOfSessions := f_NR5GC_MobileInfo_GetNoOfSessions();
    // @siclog "Step 12" siclog@
    for (i := 0; i<v_NoOfSessions; i:=i+1)
    {
      // @sic R5s210515 R5-216200 sic@
      v_PDUSessionIdToRelease := v_PDUSessionInfo[0].SessionId; // @sic R5s220614 sic@
      v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease,
                                                       cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                                                       cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00101'B)); //T3396 = 5 minutes //                                                                p_NGSM_CongestionReattemptInd,
      SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB2, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));//@sic R5s210808sic@
      SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11,
                                        tsc_NR_RbId_SRB2, //@sic R5s210808sic@
                                        cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
      v_PDN_Index := f_Get_PDN_IndexForPDUSessionId (v_PDUSessionInfo, v_PDUSessionIdToRelease); //@sic R5-220042 sic@
      if (isvalue(v_PDN_Index)) {
       f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(valueof(v_PDN_Index)));
      } else {
       f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Cannot find PDN Index of PDU Session being released");
      }
      if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
      }
    }

    //Clear stored DRB info for this session
      v_DRBIdList := f_NR_GetDRBList_ForGivenPDUSession(oct2int(v_PDUSessionIdToRelease));
     v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
      for (j:= 0; j < lengthof(v_DRBIdList); j := j+1 ) {
        v_ExistingDRBs := f_NR_DRBInfo_RemoveEntry (v_ExistingDRBs, v_DRBIdList[j]);
      }
   
    // Store updated DRB information
    f_NR5GC_MobileInfo_SetDRBInfoList(v_ExistingDRBs);  

     //Store updated PDU session information
    f_NR5GC_MobileInfo_SetSessionInfoList (v_PDUSessionInfo);

    // @siclog "Step 13" siclog@
    // Void
    
    // @siclog "Step 14" siclog@
    // The SS releases the RRC connection.
    f_NR_RRCRelease(nr_Cell11);
    
    //@siclog "Step 15" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
    f_UT_RequestIMSEmergencyCall (UT, px_EmergencyCallNumber);
    
    // @sic R5-223350 sic@
    //@siclog "Steps 16A - 16E" siclog@
    //Check: Does the UE performs steps 1-5 of Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11?
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, emergency);
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell11)), '0011'B)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
   
    // RRC Security Mode pocedure
    f_NR_RRC_ActivateSecurity (nr_Cell11, v_ReceivedMsg.SecurityProtection.NasCount);
    
    //@siclog "Step 16F" siclog@
    //The SS transmits a SERVICE ACCEPT message.
    SRB.send (cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, f_Get_NG_ServiceAcceptMsg(v_ReceivedMsg.Pdu.Msg.service_Request.pduSessionStatus))));
   
    
    //@siclog "Steps 16G-16L" siclog@
    //Check: Does the UE perform steps 8-13 of Generic Test Procedure for IMS Emergency call establishment with IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11?
    // Steps 8-13
    f_NR5GC_EmergencyCallPDUSession(nr_Cell11, tsc_NR_RbId_SRB1);
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "IMS Emergency Call Established");
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 16G-16L ");
 
    //@siclog "Step 17 - 18" siclog@
    //Make the UE release the emergency call. (NOTE 1)
    //The Generic test procedure for IMS MO Emergency call release as specified in TS 38.508-1 [4], subclause 4.9.12A takes place
    f_NR5GC_EmergencyCallRelease(nr_Cell11);

    f_NR_RRCRelease(nr_Cell11);
  } 




MCC160 Implementation

function fl_TC_11_4_4_NR5GC_TestBody() runs on NR5GC_PTC
  {


…

    // @siclog "Step 10 - 11" siclog@
    // Make the UE release the emergency call. (NOTE 1)
    f_NR5GC_EmergencyCallRelease(nr_Cell11);
    
    // @sic R5s211524 sic@

    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_NoOfSessions := f_NR5GC_MobileInfo_GetNoOfSessions();
    // @siclog "Step 12" siclog@
    for (i := 0; i<v_NoOfSessions; i:=i+1)
    {
      // @sic R5s210515 R5-216200 sic@
      v_PDUSessionIdToRelease := v_PDUSessionInfo[0].SessionId; // @sic R5s220614 sic@
      v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease,
                                                       cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                                                       cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00101'B)); //T3396 = 5 minutes //                                                                p_NGSM_CongestionReattemptInd,
      SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB2, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));//@sic R5s210808sic@
      SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11,
                                        tsc_NR_RbId_SRB2, //@sic R5s210808sic@
                                        cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
      v_PDN_Index := f_Get_PDN_IndexForPDUSessionId (v_PDUSessionInfo, v_PDUSessionIdToRelease); //@sic R5-220042 sic@
      if (isvalue(v_PDN_Index)) {
       f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(valueof(v_PDN_Index)));
      } else {
       f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Cannot find PDN Index of PDU Session being released");
      }
      if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
      }

   // MOVED }

    
      //@sic R5s220829 sic@
      //Clear stored DRB info for this session
      v_DRBIdList := f_NR_GetDRBList_ForGivenPDUSession(oct2int(v_PDUSessionIdToRelease));
      v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
      for (j:= 0; j < lengthof(v_DRBIdList); j := j+1 ) {
          v_ExistingDRBs := f_NR_DRBInfo_RemoveEntry (v_ExistingDRBs, v_DRBIdList[j]);
      }
   
      // Store updated DRB information
      f_NR5GC_MobileInfo_SetDRBInfoList(v_ExistingDRBs);  
    }

…

}


1.2 [bookmark: _Toc107494144][bookmark: _Toc107526173]Change 2
	Function name
	f_NR5GC_EmergencyCallRelease

	Reason for change
	If the UE sends a PDU SESSION RELEASE REQUEST following emergency call release, the function f_Check_NG_PDUSessionReleaseReq() is setting p_GSM_MobilityInfo.PdnIndex to omit. Since p_GSM_MobilityInfo is an out type parameter, v_PDUSession.PdnIndex is then omit when passed into f_IP_RtpRtcpLoopback(), which eventually gives a fatal error inside f_PDN_PdnIndex2Integer.

	Summary of change
	Store the PDN index to a local variable to use in f_IP_RtpRtcpLoopback(), in case it gets cleared in f_Check_NG_PDUSessionReleaseReq()

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	Accepted



Before Change:
	  function f_NR5GC_EmergencyCallRelease(NR_CellId_Type p_NR_CellId, boolean p_MOTriggerRequired := true) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN); // @sic R5s210363 sic@
    var GSM_MobilityInfo_Type v_PDUSession := valueof(v_GSM_MobilityInfo); // Must be there as previously established it
    var O1_Type v_PDUSessionId := v_PDUSession.SessionId;
    var boolean v_WaitForReq := true; // @sic R5s220524 sic@
    timer t_Wait := 5.0;
    
    if (p_MOTriggerRequired) { // @sic R5s220531 sic@
      f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]); // @sic R5s220039 sic@
    }
    t_Wait.start;
    alt {         // @sic R5-216829 sic@
      [v_WaitForReq] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp  // @sic R5s211524 sic@
      {
        if (not f_Check_NG_PDUSessionReleaseReq (v_PDUSession, f_NR5GC_MobileInfo_GetSessionInfoList(), v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId)) {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
        }
        t_Wait.stop;
        v_WaitForReq := false;
        repeat;
      }
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger)  // @sic R5s220039 sic@
         { t_Wait.stop }
      [] t_Wait.timeout   // @sic R5s220039 sic@
         { repeat }
    }

    f_NR5GC_PDUSessionRelease (p_NR_CellId,
                               v_PDUSessionId,
                               cs_GMM_GSM_Cause (omit, '00100100'B));
    f_IP_RtpRtcpLoopback(IP, v_PDUSession.PdnIndex, tsc_IMS_MediaPort_M1, omit);  // @sic R5s210363 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Emergency Call Release");
  }




After Change:
	  function f_NR5GC_EmergencyCallRelease(NR_CellId_Type p_NR_CellId, boolean p_MOTriggerRequired := true) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN); // @sic R5s210363 sic@
    var GSM_MobilityInfo_Type v_PDUSession := valueof(v_GSM_MobilityInfo); // Must be there as previously established it
    var O1_Type v_PDUSessionId := v_PDUSession.SessionId;
    var boolean v_WaitForReq := true; // @sic R5s220524 sic@
    timer t_Wait := 5.0;
    var PDN_Index_Type v_PdnIndex := v_PDUSession.PdnIndex;
    
    if (p_MOTriggerRequired) { // @sic R5s220531 sic@
      f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]); // @sic R5s220039 sic@
    }
    t_Wait.start;
    alt {         // @sic R5-216829 sic@
      [v_WaitForReq] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp  // @sic R5s211524 sic@
      {
        if (not f_Check_NG_PDUSessionReleaseReq (v_PDUSession, f_NR5GC_MobileInfo_GetSessionInfoList(), v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId)) {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
        }
        t_Wait.stop;
        v_WaitForReq := false;
        repeat;
      }
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger)  // @sic R5s220039 sic@
         { t_Wait.stop }
      [] t_Wait.timeout   // @sic R5s220039 sic@
         { repeat }
    }

    f_NR5GC_PDUSessionRelease (p_NR_CellId,
                               v_PDUSessionId,
                               cs_GMM_GSM_Cause (omit, '00100100'B));
    f_IP_RtpRtcpLoopback(IP, v_PdnIndex, tsc_IMS_MediaPort_M1, omit);  // @sic R5s210363 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Emergency Call Release");
  }




